




Programiranje kroz aplikacije

Rad sa folderima i fajlovima

Rad sa tekstualnim fajlovima

GetOpenFilename metod

- Vraća ime selektovanog fajla, uključujući i put do njega.
- U pitanju je metoda objekta `Application` koja prikazuje klasičan Open dijalog prozor (sličan onom koji se dobija opcijom File/Open), ali ne otvara odabrani fajl, već samo vraća ime fajla.
- Sintaksa ove metode je:

`Ime = Application.GetOpenFilename(FileFilter, FilterIndex, Title, ButtonText, MultiSelect)`

gdje su argumenti (svi su opcioni):

- `FileFilter` – filter fajlova, string koji određuje tipove fajlova koji će biti prikazani u padajućoj listi *Files of type* prozora Open;
- `FilterIndex` – opcija menija koja će biti podrazumijevano prikazana;
- `Title` – naslov prozora. Ukoliko se izostavi, naslov je „Open“;
- `ButtonText` – samo za Macintosh računare;
- `MultiSelect` – omogućava selekciju više fajlova (True), a ako je False (podrazumijevano), može se selektovati samo jedan fajl.

Argument FileFilter

- U pitanju je string koji određuje tipove fajlova koji će biti prikazani u padajućoj listi *Files of type* prozora Open.
- Za svaki tip fajla koji specificiramo, potrebno je navesti dva dijela odvojena zarezom. Na primjer, za .txt fajlove, filter bi mogao biti

`Filter1 = "Tekstualni fajlovi (*.txt),*.txt"`

Prvi je tekst koji se prikazuje u padajućoj listi *Files of type* (zeleno), a drugi određuje tip fajlova koji će biti prikazani u listi (crveno).

- Ukoliko želimo da prikažemo više tipova fajlova u padajućoj listi, filter bi se definisao na sljedeći način:


```
Filter2 = "Tekstualni fajlovi (*.txt),*.txt," & _  
         "Word fajlovi (*.doc),*.doc," & _  
         "Excel fajlovi (*.xls),*.xls," & _  
         "Svi fajlovi (*.*),*.*"
```

Filter2 je jedan string, prikazan u više redova radi preglednosti

Argument FileFilter - Primjeri

- Naredba `Application.GetOpenFilename(Filter1)` bi dala prozor


`Filter1 = "Tekstualni fajlovi (*.txt),*.txt"`



Argument FileFilter - Primjeri

- Naredba `Application.GetOpenFilename(Filter2, 3)` bi dala prozor

`Filter2 = "Tekstualni fajlovi (*.txt),*.txt, Word fajlovi (*.doc),*.doc," & _
"Excel fajlovi (*.xls),*.xls, Svi fajlovi (*.*),*.*"`



GetOpenFilename i GetSaveAsFilename

- **GetOpenFilename** vraća False ako korisnik pritisne dugme Cancel ili zatvori prozor pomoću dugmeta ☒ u gornjem desnom uglu.
- Slična metodi **GetOpenFilename** je metoda **GetSaveAsFilename**.
- **GetSaveAsFilename** je metoda objekta **Application** koja prikazuje Save As dijalog boks i čeka da korisnik selektuje fajl. Vraća ime fajla, uključujući i put, i ne vrši nikakvu akciju nad selektovanim fajlom.
- Za razliku od **GetOpenFilename**, ova metoda za prvi argument ima opciono **InitialFilename**, koji sugerise ime fajla. Ukoliko se izostavi, uzima se ime aktivne radne sveske. Druga razlika je da **GetSaveAsFilename** nema za posljednji argument **MultiSelect**.
- Još jednom napominjemo da ove dvije metode služe za odabir fajlova!
- Za odabir foldera je najpogodnije koristiti **FileDialog** objekt.

FileDialog objekat

- **FileDialog** objekat postoji počev od Excel-a 2002, tako da u ranijim verzijama Excel-a ovakav način odabira foldera neće raditi. Osim foldera, pomoću ovog objekta se mogu selektovati i fajlovi.
- Ovaj objekat se dobija pomoću **FileDialog** osobine objekta **Application**. Ova osobina ima jedan argument, **fileDialogType**, koji definiše tip dijaloga i ima četiri moguće vrijednosti:
 - **msoFileDialogFilePicker** – dijalog za odabir fajla;
 - **msoFileDialogFolderPicker** – dijalog za odabir foldera;
 - **msoFileDialogOpen** – dijalog za otvaranje fajla;
 - **msoFileDialogSaveAs** – dijalog za snimanje fajla.
- Dakle, kreiranje **FileDialog** objekta za odabir foldera bi bilo:
Dim FD As FileDialog
Set FD = Application.FileDialog(msoFileDialogFolderPicker)


FileDialog objekat (nastavak)

■ FileDialog objekat ima:

- osobinu **InitialFileName** koja definiše početni folder,
- metodu **Show** koja prikazuje dijalog boks,
- osobinu **SelectedItems** koja daje ime selektovanog foldera.

Primjer:

```
Dim FD As FileDialog
Set FD = Application.FileDialog(*folder*)
With FD
    .InitialFileName = "C:\Temp\"
    .Title = "Odaberite folder"
    .Show
End With
If FD.SelectedItems.Count = 0 Then
    MsgBox "Selekcija otkazana!"
Else
    MsgBox FD.SelectedItems(1)
End If
```



Rad sa fajlovima i folderima u VBA

Naredba	Opis
ChDir	Promjena tekućeg foldera.
CurDir	Dobijanje tekućeg foldera.
ChDrive	Promjena tekućeg drajva.
Dir	Dobijanje imena fajla ili foldera koje odgovara određenom obrascu, atributu fajla ili labeli drajva.
FileCopy	Kopiranje fajla.
FileDateTime	Dobijanje datuma i vremena kad je fajl posljednji put modifikovan.
FileLen	Dobijanje veličine fajla, izražene u bajtima.
GetAttr	Dobijanje atributa fajla.
Kill	Brisanje fajla.
MkDir	Kreiranje novog foldera.
Name	Promjena imena fajla ili foldera.
Rmdir	Brisanje praznog foldera.
SetAttr	Promjena atributa fajla.

Funkcija Dir

- Funkcija **Dir** vraća string koji predstavlja ime fajla ili foldera koje odgovara određenom obrascu. Njena sintaksa je:

ImeFajla = **Dir**(pathname, attributes)

- **pathname** – string koji određuje ime fajla (može uključiti i put),
 - **attributes** – VBA konstanta ili broj koji određuje atribut fajla. Na primjer, atribut **vbDirectory** određuje pretragu foldera.
- Ako **Dir** ne pronađe traženi fajl, vraća prazan string.
 - Za specificiranje više fajlova, u funkciji **Dir** se mogu koristiti džokeri:
 - * (mijenja proizvoljan broj karaktera) i
 - ? (mijenja jedan karakter).
 - Za obilazak svih fajlova u folderu, ili fajlova određenog tipa, koristi se oblik funkcije **Dir** bez argumenata, pri čemu se pri prvom pozivu ove funkcije mora specificirati put, inače će doći do greške. Kada prođe kroz sve fajlove u folderu, **Dir** vraća prazan string. U narednom pozivu funkcije **Dir** se mora navesti put da ne bi došlo do greške.

Funkcija Dir - Primjer

U aktivni radni list upisati imena svih fajlova iz određenog foldera, njihovu veličinu u bajtima, kao i datum i vrijeme posljednje modifikacije.

Za obilazak fajlova određenog tipa, potrebno je navesti tip fajla u prvom pozivu, npr.:
`f = Dir(Folder & "*.docx")`

```
Sub ListanjeFajlova()  
    Folder = "C:\Temp\  
    I = 1  
    With ActiveSheet  
        .Cells(I, 1) = "Ime fajla"  
        .Cells(I, 2) = "Veličina (bytes)"  
        .Cells(I, 3) = "Datum / Vreme"  
    End With  
    f = Dir(Folder)  
    Do While f <> ""  
        I = I + 1  
        With ActiveSheet  
            .Cells(I, 1) = f  
            .Cells(I, 2) = FileLen(Folder & f)  
            .Cells(I, 3) = FileDateTime(Folder & f)  
        End With  
        f = Dir  
    Loop  
End Sub
```

Funkcija Dir - Primjer

```
Sub ListanjeFajlova()
```

```
Folder = "C:\Temp\"
```

Početni folder

Zaglavlje
tabele

```
I = 1
```

```
With ActiveSheet
```

```
.Cells(I, 1) = "Ime fajla"
```

```
.Cells(I, 2) = "Veličina (bytes)"
```

```
.Cells(I, 3) = "Datum / Vreme"
```

```
End With
```

```
f = Dir(Folder)
```

Uzimanje prvog fajla

```
Do While f <> ""
```

```
I = I + 1
```

```
With ActiveSheet
```

```
.Cells(I, 1) = f
```

```
.Cells(I, 2) = FileLen(Folder & f)
```

```
.Cells(I, 3) = FileDateTime(Folder & f)
```

FileLen vraća veličinu fajla u bajtima

FileDateTime vraća datum i
vrijeme posljednje modifikacije.

```
End With
```

```
f = Dir
```

Uzimanje sljedećeg fajla

```
Loop
```

```
End Sub
```

Za obilazak fajlova
određenog tipa, potrebno
je navesti tip fajla u
prvom pozivu, npr.:
`f = Dir(Folder & "*.docx")`

Rad sa tekstualnim fajlovima

Otvaranje fajla

- Prije čitanja iz fajla, ili upisa u njega, fajl je potrebno otvoriti.
- Za otvaranje fajla se koristi VBA naredba **Open**. Mi ćemo koristiti pojednostavljeni oblik ove naredbe:

Open **pathname** **For** **mode** **As** **#filename**

gdje je:

- **pathname** – ime fajla, uključujući put (obavezan argument);
- **mode** – način otvaranja fajla (obavezan argument) i može biti:
 - **Append** – čitanje fajla ili dopisivanje teksta na kraj fajla;
 - **Input** – čitanje fajla, bez mogućnosti upisa;
 - **Output** – čitanje i upis u fajl. Na ovaj način se kreiraju novi fajlovi, dok se brišu postojeći fajlovi sa istim imenom.
- **filename** – broj fajla (obavezan argument). U pitanju je broj između **1** i **511**; još se naziva i *handle fajla*. Radi dobijanja broja fajla koji je dostupan (trenutno nije u upotrebi), može se koristiti funkcija **FreeFile**.

Otvaranje tekst fajlova - Primjeri

- Otvaranje fajla C:\Temp\Fajl.txt radi čitanja iz njega se vrši na sljedeći način:

```
Open "C:\Temp\Fajl.txt" For Input As #1
```

- Ako ne znamo broj fajla koji je dostupan, možemo koristiti sljedeće:

```
Dim Dostupan As Integer
```

```
Dostupan = FreeFile
```

```
Open "C:\Temp\Fajl.txt" For Input As #Dostupan
```

- Otvaranje fajla C:\Temp\Fajl.txt radi upisa u njega se vrši na sljedeći način:

```
Open "C:\Temp\Fajl.txt" For Output As #1
```

Zatvaranje tekst fajlova

- Fajl koji je otvoren naredbom **Open** se zatvara naredbom **Close**, u sljedećem zapisu:
Close #filenumber
- Ovom se naredbom može zatvoriti više fajlova, na sljedeći način:
Close #filenumber1, #filenumber2, ..., #filenumberN
- Ukoliko se navede samo naredba **Close**, bez navođenja broja fajlova, zatvaraju se svi trenutno otvoreni fajlovi.
- Svi fajlovi otvoreni naredbom **Open** se alternativno mogu zatvoriti naredbom **Reset**.

Čitanje iz tekst fajlova

- Kod VBA postoje tri naredbe za čitanje podataka iz tekst fajlova:

- **Input** – funkcija za čitanje određenog broja karaktera iz fajla;
- **Input #** – čitanje podataka iz fajla i dodeljivanje tih podataka promenljivim koje su odvojene zarezima;
- **Line Input #** – čitanje čitave linije podataka.

- Funkcija **Input** čita određeni broj karaktera iz fajla i vraća string pročitanih karaktera. Poziva se na sljedeći način:

Procitano = **Input**(number, #filename)

gdje je **number** broj karaktera koji se čita, a **filename** je handle za dati fajl. Na primjer, sekvenca naredbi

Open "C:\temp\Fajl.txt" For Input As #1

Procitano = Input(12, #1)

će pročitati prvih 12 karaktera iz fajla Fajl.txt i upisati ih u promenljivu **Procitano**.

Funkcija Input - Primjer

- Napisati proceduru koja broji koliko ima cifara u fajlu Fajl.txt. Dobijeni broj prikazati pomoću message boksa.

```
Sub BrojCifaraUFajlu()  
    Dim BrCif As Integer, S As String  
    Open "C:\Temp\Fajl.txt" For Input As #1  
    BrCif = 0  
    Do Until EOF(1) ←  
        S = Input(1, #1)  
        If S Like "[0-9]" Then  
            BrCif = BrCif + 1  
        End If  
    Loop  
    MsgBox "Ima " & BrCif & " cifara"  
    Close #1  
End Sub
```

Funkcija **EOF** vraća True kada se dođe do kraja fajla i False u suprotnom.
Argument ove funkcije je handle na otvoreni fajl, naveden bez #.
Ako ne koristimo **EOF**, dolazi do greške prilikom pokušaja čitanja nakon posljednjeg karaktera.

Naredba Input

- Naredba **Input #** je pogodna ukoliko svaki red fajla sadrži fiksni broj podataka.
- **Input #** čita podatke iz otvorenog fajla i dodeljuje ih promenljivim koje su odvojene zarezima.
- Na primjer, pretpostavimo da jedan red fajla čiji je handle **#1** sadrži string (naveden po navodnicima) i cio broj (npr. "**Novembar**", 8). Učitavanje ova dva podatka u promenljive **S** i **X** se vrši kao:
Input #1, S, X
- Ilustrujmo naredbu na sljedećem primjeru. Pretpostavimo da svaki red fajla **Ispit.txt** sadrži ime i prezime studenta i broj poena koje je taj student dobio na ispitu. Ime i prezime su zadati u formi jednog stringa pod znacima navoda, dok je broj poena realan broj. Student je položio ispit ukoliko je dobio više od 50 poena. Napisati proceduru koja određuje i ispisuje koliko je studenata položilo ispit.

Naredba Input # - Primjer

```
Sub Polozili()  
    Dim ImePrez As String, Poeni As Single, BrPol As Integer  
    Open "C:\Temp\Ispit.txt" For Input As #1  
    BrPol = 0  
    Do Until EOF(1)  
        Input #1, ImePrez, Poeni  
        If Poeni > 50 Then BrPol = BrPol + 1  
    Loop  
    MsgBox "Položilo je " & BrPol & " studenata"  
    Close #1  
End Sub
```

Naredba Line Input

- Naredba **Line Input #** čita jednu liniju teksta (do znaka za novi red) i pročitano liniju smješta u **String** promenljivu.
- Sintaksa ove naredbe je:
Line Input #filename, varname
gdje je **varname** ime promjenljive u koju smještamo pročitano liniju.
- Dalji rad se svodi na obradu stringova.

Upis u tekst fajl. Naredba Write

- Za upis u tekstualni fajl koristimo naredbe **Write #** i **Print #**.
- Naredba **Write #** služi za upis liste vrijednosti. **Upisane vrijednosti su razdvojene zarezima u fajlu, dok se stringovi nalaze pod znacima navoda.**
- Podaci upisani u fajl sa **Write #** se obično čitaju naredbom **Input #** (sjetite se znaka navoda pri čitanju stringova sa **Input #**).
- Sintaksa naredbe **Write #** je:
Write #filename, outputlist
gdje je **outputlist** lista promenljivih razdvojenih zarezima koje se upisuju u fajl. Umesto promenljive se može navesti i konstantna vrednost, bilo numerička, stringovna ili logička. Ukoliko se **outputlist** izostavi, upisuje se prazna linija u fajl.
- Ukoliko želimo da nakon ispisa ostanemo u istom redu, naredbu **Write #** treba završiti tačka-zarezom.

Naredba Print

- Naredba **Print #** takođe služi za upis liste vrijednosti i njena sintaksa je:

Print #filename, outputlist

gdje je **outputlist** lista vrijednosti koje želimo da upišemo u fajl. **Ukoliko outputlist ima više vrijednosti, te vrijednosti treba razdvajati tačka-zarezom. Ako se razdvajaju zarezima, u fajlu će odgovarajuće vrijednosti biti razdvojene tabovima.**

- Za razliku od **Write #**, tekst je upisan onako kako smo mi naveli, bez dodavanja znaka navoda oko stringova i zareza koji razdvajaju pojedinačne unose. Ukoliko se **outputlist** izostavi, upisuje se prazna linija u fajl.
- Podaci upisani u fajl naredbom **Print #** se obično čitaju naredbama **Input** i **Line Input #**.
- Kao kod naredbe **Write #**, ako želimo da nakon ispisa ostanemo u istom redu, naredbu **Print #** treba završiti tačka-zarezom.

Rad sa tekst fajlovima - Primjeri

- Napisati proceduru koja otvara tekstualni fajl C:\Fajl.txt i sve linije fajla koje sadrže riječ "VBA" prepisuje u radni list. Pored pronađene linije, u radni list upisati i redni broj linije i njenu dužinu.

```
Sub Podstring()  
    Dim I As Integer, BrLin As Integer  
    Dim Linija As String  
    Open "C:\Fajl.txt" For Input As #1  
    Cells(1, 1) = "Linija"  
    Cells(1, 2) = "Broj linije"  
    Cells(1, 3) = "Dužina linije"  
    Range("A1:C1").Font.Bold = True  
    I = 1  
    BrLin = 0
```

```
Do Until EOF(1)  
    Line Input #1, Linija  
    BrLin = BrLin + 1  
    If InStr(Linija, "VBA") <> 0 Then  
        I = I + 1  
        Cells(I, 1) = Linija  
        Cells(I, 2) = BrLin  
        Cells(I, 3) = Len(Linija)  
    End If  
Loop  
Close #1  
End Sub
```


Rad sa tekst fajlovima - Primjeri

- Formirati fajl **Vrijednosti.txt** koji sadrži numeričke vrijednosti ćelija selektovanog opsega (svaka u zasebnom redu fajla) u formatu:

vrsta: I, kolona: J, vrednost: X

gdje je X vrijednost ćelije u I-toj vrsti i J-toj koloni selektovanog opsega. Ne upisivati prazne ćelije i one koje ne sadrže broj.

```
Sub Vrednosti()  
    Dim I As Integer, J As Integer, Cel As Variant  
    Open "C:\Temp\Vrijednosti.txt" For Output As #1  
    For I = 1 To Selection.Rows.Count  
        For J = 1 To Selection.Columns.Count  
            Cel = Selection.Cells(I, J)  
            If Cel <> "" And IsNumeric(Cel) Then  
                Print #1, "vrsta: " & I & ", kolona: " & J & ", vrednost: " & Cel  
            End If  
        Next  
    Next  
    Close #1  
End Sub
```



Šta bi se upisalo u fajl da je
umjesto Print korišćeno Write?